POGLAVJE II

II.7 Statična upogibnica kot interpolacijska funkcija

1. Izračun statične upogibnice za izbrano obtežbo
2. Izračun posplošene mase, posplošene togosti in posplošene geometrijske togosti
3. Izračun lastne frekvence ter kritične uklonske sile

1. Izračun statične upogibnice za izbrano obtežbo

V prejšnjem podpoglavju je bilo pokazano, da se lahko do inženirsko uporabnih vrednosti za približek prve lastne frekvence in/ali kritične uklonske sile pride tudi brez reševanja navadne diferencialne enačbe, če se uporabijo ustrezne interpolacijske funkcije (v obliki polinomov ali trigonometričnih vrst ali kombinacije), ki se poiščejo glede na robne pogoje (ali pa tudi pogoje zveznosti) obravnavane konstrukcije.

Kadar pa je obravnavana konstrukcija relativno enostavna (npr. en sam element), pa je zanjo smiselno uporabiti kar »naravne« interpolacijske funkcije, ki seveda avtomatično ustrezajo robnim pogojem - to so upogibnice zaradi različnih statičnih prečnih obtežb. Mogoče jih je najti v raznih tabelah, ali pa se za obravnavano konstrukcijo lahko tudi izpeljejo z zaporednim integriranjem obtežbe, ali pa če je konstrukcija statično določena, kar z integracijo momentov.

Primer bo tako demonstriral uporabo statične upogibnice za vajo 10 (http://www.geocities.com/mcsDISK/e10.pdf), ki obravnava steber – konzolno konstrukcijo in zato datoteki damo naslov Vaja 10. Prva faza analize pa obravnava statično upogibnico zaradi enakomerne zvezne obtežbe *q*. Zato se definira obtežba kot:

Vaja 10

■ Staticna upogibnica zaradi enakomerne zvezne obtezbe

```
In[1]:= obtezba = q;
```

Prečne sile sledijo z integracijo prečne obtežbe, doda pa se še integracijska konstanta:

```
In[2]:= V[x_] = Integrate[obtezba, x] + C1
```

Out[2]= $C1 + q\,x$

Ker obstaja informacija o prečni sili na prostem koncu konzole, je mogoče neznano konstanto določiti kar takoj, mogoče pa je vse konstante določiti tudi na koncu analize pomikov, po zadnji integraciji (kot bo storjeno tukaj). Zato se prečne sile integrirajo, da se pridobi izraz za upogibni moment:

```
In[3]:= M[x_] = Integrate[V[x], x] + C2
```

Out[3]= $C2 + C1\,x + \frac{q\,x^2}{2}$

Pri naslednji integraciji, ki vodi do zasuka φ, je potrebno integral najprej deliti še z upogibno togostjo EI (kar je v obravnavanem primeru, ko je prerez konstanten, material pa homogen, sicer mogoče storiti tudi po integraciji):

```
In[4]:= φ[x_] = Integrate[M[x] / EI, x] + C3
```

$$\text{Out[4]=}\quad C3+\frac{C2\,x+\frac{C1\,x^2}{2}+\frac{q\,x^3}{6}}{EI}$$

Končno sledi enačba upogibnice oz. prečnega pomika po integraciji zasuka φ:

```
In[5]:= v[x_] = Integrate[φ[x], x] + C4
```

$$\text{Out[5]=}\quad C4+C3\,x+\frac{C2\,x^2}{2\,EI}+\frac{C1\,x^3}{6\,EI}+\frac{q\,x^4}{24\,EI}$$

Dobljena zveza velja za poljubno konstrukcijo oz. robne pogoje. Ker pa je v obravnavanem primeru konstrukcija konzola, se upoštevajo ustrezni robni pogoji (koordinata x npr. narašča od dna stebra proti vrhu). Zapišejo se v obliki štirih linearnih enačb s štirimi neznankami (C1, C2, C3 in C4), ki se rešijo z ukazom *Solve[{enačbe},{neznanke}]*. Vse rešitve se shranijo npr. v spremenljivko (vektor) *konstante*.

■ Dolocitev konstant za konzolo

```
In[6]:= konstante = Solve[{V[L] == 0, M[L] == 0, φ[0] == 0, v[0] == 0},
          {C1, C2, C3, C4}]
```

$$\text{Out[6]=}\quad \left\{\left\{C2\to\frac{L^2\,q}{2},\ C3\to 0,\ C4\to 0,\ C1\to -L\,q\right\}\right\}$$

Konstantam C1, C2, C3 in C4 je sedaj potrebno še dejansko dodeliti (predlagane) vrednosti, izračunane s sistemom enačb (in shranjene v spremenljivki *konstante*):

```
In[7]:= C1 = C1 /. konstante[[1]]
        C2 = C2 /. konstante[[1]]
        C3 = C3 /. konstante[[1]]
        C4 = C4 /. konstante[[1]]
```

$\text{Out[7]=}\quad -L\,q$

$$\text{Out[8]=}\quad \frac{L^2\,q}{2}$$

$\text{Out[9]=}\quad 0$

$\text{Out[10]=}\quad 0$

Izpis bi bil mogoč tudi brez izhodnih vrstic, če bi se na koncu vsake vhodne vrstice zapisalo podpičje. Iskana upogibnica je tako:

```
In[11]:= v[x]
```

Out[11]= $\frac{L^2 q x^2}{4 EI} - \frac{L q x^3}{6 EI} + \frac{q x^4}{24 EI}$

2. Izračun posplošene mase, posplošene togosti in posplošene geometrijske togosti

Sedaj se je mogoče lotiti izračuna posplošenih dinamičnih parametrov. Posplošena masa *m** oz. *mz* je tako:

■ Dinamicni parametri

```
In[12]:= mz = Integrate[m v[x]^2, {x, 0, L}]
```

Out[12]= $\frac{13 L^9 m q^2}{3240 EI^2}$

posplošena togost pa k^* oz. *kz*:

```
In[13]:= kz = Integrate[ EI v''[x]^2, {x, 0, L}]
```

Out[13]= $\frac{L^5 q^2}{20 EI}$

Če se želi upoštevati še osna sila zaradi lastne teže konstrukcije, se le ta definira kot (opomba: osne sile ni mogoče definirati kot N[x], saj je velika črka N rezervirana za ukaz za numerični izračun parametra). Ker je jasno, da bo osna sila linearno upadala od dna stebra, kjer je enaka $m \cdot g \cdot L$, proti vrhu stebra, kjer je enaka nič, se lahko kar izračuna z ukazom za interpolacijo polinomov *InterpolatingPolynomial[{diskretne točke in pripadajoče vrednosti funkcije},parameter]*:

```
In[14]:= osnasila[x_] = InterpolatingPolynomial[
           {{0, m g L}, {L, 0}}, x]
```

Out[14]= $g L m - g m x$

Posplošena geometrijska togost zaradi lastne teže konstrukcije je tako:

```
In[15]:= kglt = Integrate[osnasila[x] v'[x]^2, {x, 0, L}]
```

Out[15]= $\frac{g L^8 m q^2}{160 EI^2}$

Če pa se želi poiskati posplošena geometrijska togost zaradi koncentrirane vertikalne sile P_{ukl} na vrhu konstrukcije, ki povzroča konstantno tlačno osno silo P_{ukl}, pa sledi:

In[16]:= `kgP = Integrate[Pukl v'[x]^2, {x, 0, L}]`

Out[16]= $\frac{L^7\,\text{Pukl}\,q^2}{56\,\text{EI}^2}$

Lastna frekvenca (brez upoštevanja geometrijske togosti) je tako:

In[17]:= $\omega = \sqrt{\frac{\text{kz}}{\text{mz}}}$

Out[17]= $9\sqrt{\frac{2}{13}}\sqrt{\frac{\text{EI}}{L^4\,m}}$

oziroma:

In[18]:= `N[%]`

Out[18]= $3.53009\sqrt{\frac{\text{EI}}{L^4\,m}}$

Lastna frekvenca (ob upoštevanju geometrijske togosti zaradi lastne teže) pa je:

In[19]:= $\omega = \text{Simplify}\left[\sqrt{\frac{\text{kz} - \text{kglt}}{\text{mz}}}\right]$

Out[19]= $\frac{9\sqrt{-\frac{9}{L} + \frac{8\,\text{EI}}{L^4\,m}}}{2\sqrt{13}}$

Kritična uklonska sila P_{ukl} pa se izračuna iz že znanega pogoja, da sta posplošena togost in posplošena geometrijska togost enaki (zanemarjen je del geometrijske togosti, ki pripada lastni teži):

In[20]:= `Solve[kz - kgP == 0, Pukl]`

Out[20]= $\left\{\left\{\text{Pukl} \to \frac{14\,\text{EI}}{5\,L^2}\right\}\right\}$

oziroma:

In[21]:= `N[%]`

Out[21]= $\left\{\left\{\text{Pukl} \to \frac{2.8\,\text{EI}}{L^2}\right\}\right\}$

Pripomniti je morda še potrebno, da se zgolj v primerih, ko je kot interpolacijska funkcija izbrana upogibnica zaradi statične obtežbe, posplošena togost lahko alternativno izračuna tudi s pomočjo dela obtežbe na upogibnici:

In[22]:= `kzalt = Integrate[q v[x], {x, 0, L}]`

Out[22]= $\frac{L^5 q^2}{20\,EI}$

kar seveda vodi do enakega rezultata. Ker *Mathematica* običajno zlahka zmore izvrednotiti integral, ki nastopa pri splošnem izrazu za posplošeno togost, uporaba tega izraza ne predstavlja takšne prednosti kot v primeru »peš« analize problema.

Ker pa je konstrukcija statično določena, je mogoče enostavno poiskati diagram momentov in kar zgolj z dvakratno integracijo priti do upogibnice. Za konzolo, na vrhu obremenjeno s koncentrirano prečno silo P, tako velja:

■ Staticna upogibnica zaradi sile P na vrhu

In[23]:= `M[x_] = P (L - x)`

Out[23]= $P\,(L - x)$

Integracija momenta seveda vodi do zasuka φ:

In[24]:= `φ[x_] = Integrate[M[x] / EI, x] + C5`

Out[24]= $C5 + \frac{P\left(L\,x - \frac{x^2}{2}\right)}{EI}$

ponovna integracija pa do upogibnice:

In[25]:= `v[x_] = Integrate[φ[x], x] + C6`

Out[25]= $C6 + C5\,x + \frac{L\,P\,x^2}{2\,EI} - \frac{P\,x^3}{6\,EI}$

za katero je potrebno še določiti konstanti na osnovi (zgolj) dveh robnih pogojev:

In[26]:= `konstante = Solve[{φ[0] == 0, v[0] == 0}, {C5, C6}]`

Out[26]= $\{\{C5 \to 0,\ C6 \to 0\}\}$

Izračunane vrednosti konstant se lahko upoštevajo v funkciji pomika kar direktno brez definiranja vrednosti posamezne konstante:

In[27]:= `v[x_] = v[x] /. konstante[[1]]`

Out[27]= $\frac{L\,P\,x^2}{2\,EI} - \frac{P\,x^3}{6\,EI}$

kar omogoči izračun posplošene mase:

```
In[28]:= mz = Integrate[m v[x]^2, {x, 0, L}]
```

Out[28]= $\frac{11\,L^7\,m\,P^2}{420\,EI^2}$

kot tudi posplošene togosti (v primeru koncentrirane obtežbe celo brez integriranja):

```
In[29]:= kz = P v[L]
```

Out[29]= $\frac{L^3\,P^2}{3\,EI}$

Posplošena geometrijska togost pa znaša:

```
In[30]:= kgP = Integrate[Pukl v'[x]^2, {x, 0, L}]
```

Out[30]= $\frac{2\,L^5\,P^2\,Pukl}{15\,EI^2}$

Približek prve lastne frekvence je tako:

In[31]:= $\omega = N\left[\sqrt{\frac{kz}{mz}}\right]$

Out[31]= $3.56753\,\sqrt{\frac{EI}{L^4\,m}}$

približek kritične uklonske sile pa znaša:

```
In[32]:= N[Solve[kz - kgP == 0, Pukl]]
```

Out[32]= $\left\{\left\{Pukl \to \frac{2.5\,EI}{L^2}\right\}\right\}$

NALOGE ZA SAMOSTOJNO DELO

1. Poišči lastno frekvenco ter kritično uklonsko silo zaradi trikotne zvezne obtežbe, ki ima maksimum na dnu stebra.
2. Poišči lastno frekvenco ter kritično uklonsko silo za primer trikotne zvezne obtežbe, ki ima maksimum na vrhu stebra.
3. Poišči lastno frekvenco ter kritično uklonsko silo zaradi koncentriranega momenta na vrhu stebra.

Katera izmed dobljenih rešitev vodi do najboljšega in katera do najslabšega približka?

REŠITVE

1. $\omega = 3.63318042491699 \cdot \sqrt{\frac{EI}{L^4 \cdot m}}$ $\qquad P_{ukl} = 3.2142857142857144 \cdot \frac{EI}{L^2}$
2. $\omega = 3.5171457568454385 \cdot \sqrt{\frac{EI}{L^4 \cdot m}}$ $\qquad P_{ukl} = 2.69540555870675 \cdot \frac{EI}{L^2}$
3. $\omega = 4.47213595499958 \cdot \sqrt{\frac{EI}{L^4 \cdot m}}$ $\qquad P_{ukl} = 3 \cdot \frac{EI}{L^2}$